NAG Library Routine Document

S18GKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

S18GKF returns a sequence of values for the Bessel functions $J_{\alpha+n-1}(z)$ or $J_{\alpha-n+1}(z)$ for complex z, non-negative $\alpha < 1$ and $n = 1, 2, \ldots, |N| + 1$.

2 Specification

```
SUBROUTINE S18GKF (Z, A, NL, B, IFAIL)
INTEGER                 NL, IFAIL
REAL (KIND=nag_wp)      A
COMPLEX (KIND=nag_wp) Z, B(abs(NL)+1)
```

3 Description

S18GKF evaluates a sequence of values for the Bessel function of the first kind $J_\alpha(z)$, where z is complex and nonzero and α is the order with $0 \leq \alpha < 1$. The $(|N| + 1)$-member sequence is generated for orders $\alpha, \alpha + 1, \ldots, \alpha + |N|$ when $N \geq 0$. Note that $+$ is replaced by $-$ when $N < 0$. For positive orders the routine may also be called with $z = 0$, since $J_q(0) = 0$ when $q > 0$. For negative orders the formula

$$J_{-q}(z) = \cos(\pi q) J_q(z) - \sin(\pi q) Y_q(z)$$

is used to generate the required sequence. The appropriate values of $J_q(z)$ and $Y_q(z)$ are obtained by calls to S17DCF and S17DEF.

4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

5 Arguments

1: Z – COMPLEX (KIND=nag_wp) *Input*

On entry: the argument z of the function.

Constraint: Z $\neq (0.0, 0.0)$ when NL < 0.

2: A – REAL (KIND=nag_wp) *Input*

On entry: the order α of the first member in the required sequence of function values.

Constraint: $0.0 \leq$ A < 1.0.

3: NL – INTEGER *Input*

On entry: the value of N.

Constraint: abs(NL) ≤ 101.

4: B(abs(NL) + 1) – COMPLEX (KIND=nag_wp) array *Output*

On exit: with IFAIL $= 0$ or 3, the required sequence of function values: B(n) contains $J_{\alpha+n-1}(z)$ if NL ≥ 0 and $J_{\alpha-n+1}(z)$ otherwise, for $n = 1, 2, \ldots,$ abs(NL) + 1.

5: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, Z $= (0.0, 0.0)$ when NL < 0,
or A < 0.0,
or A ≥ 1.0,
or abs(NL) > 101.

IFAIL $= 2$

The computation has been abandoned due to the likelihood of overflow.

IFAIL $= 3$

The computation has been completed but some precision has been lost.

IFAIL $= 4$

The computation has been abandoned due to total loss of precision.

IFAIL $= 5$

The computation has been abandoned due to failure to satisfy the termination condition.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

All constants in S17DCF and S17DEF are specified to approximately 18 digits of precision. If t denotes the number of digits of precision in the floating-point arithmetic being used, then clearly the maximum number of correct digits in the results obtained is limited by $p = \min(t, 18)$. Because of errors in argument reduction when computing elementary functions inside S17DCF and S17DEF, the actual number of correct digits is limited, in general, by $p - s$, where $s \approx \max(1, |\log_{10}|z||, |\log_{10}|\alpha||)$ represents the number of digits lost due to the argument reduction. Thus the larger the values of $|z|$ and $|\alpha|$, the less the precision in the result.

8 Parallelism and Performance

S18GKF is not threaded in any implementation.

9 Further Comments

None.

10 Example

This example evaluates $J_0(z), J_1(z), J_2(z)$ and $J_3(z)$ at $z = 0.6 - 0.8i$, and prints the results.

10.1 Program Text

```
    Program s18gkfe

!     S18GKF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, s18gkf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Complex (Kind=nag_wp)            :: z
      Real (Kind=nag_wp)               :: a, alpha
      Integer                          :: i, ifail, nl
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: b(:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: abs, real, sign
!     .. Executable Statements ..
      Write (nout,*) 'S18GKF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) z, a, nl
      Allocate (b(abs(nl)+1))

      ifail = 0
      Call s18gkf(z,a,nl,b,ifail)

      Write (nout,*)
      Write (nout,*) '      Z           A     NL'
      Write (nout,*)
      Write (nout,99999) z, a, nl

      Write (nout,*)
      Write (nout,*) ' Requested values of J_alpha(Z)'
      Write (nout,*)
      Write (nout,*) '     alpha              J_alpha(Z)'
```

```
      alpha = a

      Do i = 1, abs(nl) + 1
        Write (nout,99998) alpha, b(i)
        alpha = alpha + sign(1.0E0_nag_wp,real(nl,kind=nag_wp))
      End Do

99999 Format (1X,'( ',F4.1,', ',F4.1,' )',2X,F4.1,I6)
99998 Format (1X,1P,E12.4,3X,'(',E12.4,',',E12.4,' )')
    End Program s18gkfe
```

10.2 Program Data

```
S18GKF Example Program Data
 ( 0.6,-0.8)   0.0   3 : Values of Z, A and NL
```

10.3 Program Results

```
 S18GKF Example Program Results

       Z           A     NL

 (  0.6, -0.8 )   0.0     3

  Requested values of J_alpha(Z)

      alpha              J_alpha(Z)
   0.0000E+00   (  1.0565E+00,  2.4811E-01 )
   1.0000E+00   (  3.5825E-01, -3.7539E-01 )
   2.0000E+00   ( -2.5974E-02, -1.2538E-01 )
   3.0000E+00   ( -1.9369E-02, -8.6380E-03 )
```